

This document was prepared in conjunction with work accomplished under Contract No. DE-AC09-96SR18500 with the U. S. Department of Energy.

DISCLAIMER

This report was prepared as an account of work sponsored by an agency of the United States Government. Neither the United States Government nor any agency thereof, nor any of their employees, makes any warranty, express or implied, or assumes any legal liability or responsibility for the accuracy, completeness, or usefulness of any information, apparatus, product or process disclosed, or represents that its use would not infringe privately owned rights. Reference herein to any specific commercial product, process or service by trade name, trademark, manufacturer, or otherwise does not necessarily constitute or imply its endorsement, recommendation, or favoring by the United States Government or any agency thereof. The views and opinions of authors expressed herein do not necessarily state or reflect those of the United States Government or any agency thereof.

This report has been reproduced directly from the best available copy.

**Available for sale to the public, in paper, from: U.S. Department of Commerce, National Technical Information Service, 5285 Port Royal Road, Springfield, VA 22161,
phone: (800) 553-6847,
fax: (703) 605-6900
email: orders@ntis.fedworld.gov
online ordering: <http://www.ntis.gov/help/index.asp>**

**Available electronically at <http://www.osti.gov/bridge>
Available for a processing fee to U.S. Department of Energy and its contractors, in paper, from: U.S. Department of Energy, Office of Scientific and Technical Information, P.O. Box 62, Oak Ridge, TN 37831-0062,
phone: (865)576-8401,
fax: (865)576-5728
email: reports@adonis.osti.gov**

A DYNAMIC MODEL OF THE ACTINIDE REMOVAL PROCESS AT THE SAVANNAH RIVER SITE

by

T. Hang
Westinghouse Savannah River Company
Savannah River Technology Center
Aiken, SC 29808

A paper proposed for presentation at the
SCS Advanced Simulation Technologies Conference
Arlington, Virginia
April 18 -22, 2004

and for publication in the proceedings of the meeting

This paper was prepared in connection with work done under Contract No. DE-AC09 96SR18500 with the U. S. Department of Energy. By acceptance of this paper, the publisher and/or recipient acknowledges the U. S. Government's right to retain a nonexclusive, royalty-free license in and to any copyright covering this paper, along with the right to reproduce and to authorize others to reproduce all or part of the copyrighted paper.

A Dynamic Model Of The Actinide Removal Process At The Savannah River Site

T. Hang
Westinghouse Savannah River Company
Savannah River Technology Center
Aiken, SC 29808
thong.hang@srs.gov

Keywords: Environmental science, industrial processes, waste processing technology

ABSTRACT

The Actinide Removal Process (ARP) is used at the Savannah River Site (SRS) to remove strontium, plutonium, neptunium and uranium present in the waste salt solution to meet regulatory specifications. Strontium and the actinide radioactive nuclides are removed from the salt solution by adsorption onto monosodium titanate (MST). A detailed and complete dynamic model of the ARP has been developed using Aspen Custom Modeler™ (ACM). The model simulates the ARP batch operations and computes the total cycle time for any given specific waste feed to be processed.

This paper focuses on the model development and provides the result of a typical simulation scenario.

BACKGROUND

The High Level Waste (HLW) system at the SRS processes and converts high level wastes to glass. The system can be divided by function into operational sub-systems: (1) Waste storage in the Tank Farm, (2) Waste pretreatment and removal in sludge and salt processing facilities, and (3) Waste final processing by vitrification in the Defense Waste Processing Facility (DWPF) and by disposition as grout in Saltstone vaults.

In the second sub-system, prior to the salt processing, the waste solution is pretreated in the ARP facility to remove strontium, plutonium, neptunium and uranium by adsorption onto MST. Strontium and the actinides must be sufficiently removed to meet Saltstone specifications.

Figure 1 shows a simplified flow diagram of the ARP. The Precipitate Tank (PT) is the ARP main unit. The PT process heel is maintained at 950 gallons. Four operations are performed in the PT:

1. Dilution of the salt waste.
2. Chemical treatment (adsorption).

3. Solids concentration.
4. Solids washing.




Figure 1. ARP flow diagram

The salt solution is transferred to the PT at a high sodium concentration (6-7 M). Before pretreatment to remove actinides, the solution must be diluted with 1.66 M caustic to reduce sodium concentration to ~5.6 M.

After reaching the target sodium concentration, the salt waste is treated with ~15 Wt% MST slurry. The desired concentration of MST for each batch in the PT is 0.4 g of MST per liter of total slurry. The MST solid particles are contacted with the 5.6 M salt solution to adsorb strontium and actinides. A reaction time of 24 hours is taken in order to provide ample time for the adsorption equilibrium to be achieved.

Once strontium and actinides have been adsorbed onto the MST solids, they can be separated from the salt solution along with the entrained sludge solids by filtration. During filtration, to prevent filter blocking and clogging with solids

buildup, 140 gallons of de-ionized (DI) water are used to backpulse the filter. Two backpulses are assumed for every filter batch. Clarified salt solution (i.e., the filtrate from the filtration process) is lower in radioactivity than the salt waste feed because strontium, actinides, and entrained sludge solids have been filtered out. Filtrate is stored in the Filtrate Hold Tank (FHT) to be either transferred to Saltstone or further processed in the downstream Salt Processing facility.

Several batches of salt solution are processed until the concentration of solids in the PT process heel is high enough (~5 Wt%) to be transferred to DWPF via the Low Point Pump Pit Precipitate Tank (LPPPPT). When the desired solids concentration has been reached, the process heel still contains a large amount of sodium and various salts. The heel must be washed to lower the sodium concentration to ~0.5 M prior to transfer to DWPF for vitrification. The washing is done by adding de-ionized (DI) water to the heel. The amount of water addition depends on the process heel sodium concentration and the volume of salt solution in the process heel.

After solids washing, the filter must be cleaned with chemicals. The filter cleaning is a five-step batch process.

| Addition | Transfer Time (hrs) | Cleaning Time (hr) | Total Time (hrs) |
|------------------------------|---------------------|--------------------|------------------|
| 1M caustic (500 gals) | 2 | 1 | 3 |
| DI water (500 gals) | 1 | 1 | 2 |
| 2 Wt% oxalic acid (500 gals) | 1 | 1 | 2 |
| DI water (500 gals) | 1 | 1 | 2 |
| 1M caustic (500 gals) | 2 | 1 | 3 |
| | | | $\Sigma = 12$ |

Each 500-gal chemical batch is circulated through the filter for approximately one hour for cleaning. The spent filter cleaning chemicals are fed to the PT after circulation is complete. The total volume sent to the PT is 2,500 gallons. When the cleaning caustic and oxalic acid are combined in the PT, both react to form sodium oxalate and water. The entire cleaning process takes 12 hours.

Upon completion of filter cleaning, the transfer of the spent filter cleaning chemicals to the LPPPPT functions as a line flush for the previous heel transfer.

MODEL DEVELOPMENT

ACM Modeling

The Aspen Custom Modeler™ (ACM) marketed by Aspen Technology, Inc. has been selected as the process simulation platform to develop the ARP model. The architecture of ACM makes it well suited to modeling both continuous and/or batch operations. Complex ACM models have been successfully developed by SRS researchers in the past [Hang and Walker 2001, Hang et al. 2003, Smith 2003]. Physical properties of compounds and mixtures in the ARP model are predicted using Aspen Properties Plus®.

ARP Flowsheet

Figure 2 displays the features of the ARP flowsheet, which are included in the current ACM model. The ARP is primarily described by the *ARP* block with streams representing inputs/outputs to and from the process. The *ARP* block combines both the PT and filter operations. The *ARP_FeedAdjust* performs the calculation of steady-state mixing of the salt waste solution with caustic and MST slurry to satisfy the requirements of 5.6 M sodium and 0.4 g of MST per liter of total slurry.




Figure 2. Currently modeled ARP flowsheet

Chemical Compounds

The 14 chemical compounds handled by the current evaporator models are: H_2O , Na_2CO_3 , Na_2SiO_3 , Na_2SO_4 , Na_3PO_4 , NaAlO_2 , NaCl , NaF , NaNO_2 , NaNO_3 , NaOH , MST ($\text{Na}_2\text{Ti}_2\text{O}_5\text{H}$), $\text{Na}_2\text{C}_2\text{O}_4$ and $\text{C}_2\text{H}_2\text{O}_4$. In addition, Cs, Pu, and U in the form of soluble and solid trace components are used to track radionuclides present in the system.

ARP Operation

The ARP is a batch operation. The sequence of the ARP operation is outlined in the following steps:

1. Transfer 4000 gallons of salt waste solution at 40 gpm to the PT.
2. Dilute the salt waste with 1.66 M NaOH + 15 Wt% MST at 40 gpm to achieve ~5.6M sodium slurry. The ~5.6M sodium concentration includes the incoming salt solution, caustic solution and MST slurry. The desired concentration of MST per batch in the PT is 0.4 g of MST per liter of total slurry. Steps 1 + 2 are assumed to take approximately 2 hours.
3. 24-hour reaction time for MST to adsorb strontium and actinides.
4. Concentrate salt solution to the PT process heel (950 gallons). Salt solution is fed to the cross flow filter at 850 gpm. Filtrate flows to the FHT at 4.6 gpm, and the solids go back to the PT. During filtration, the filter is backpulsed with 140 gallons of process water twice. Each filter backpulse is assumed to last 2 seconds. When backpulse is on, the backpulse water and filter feed are sent back to the PT. Steps 1 to 4 constitute a filter batch.
5. Process many filter batches until the solids concentration in the PT process heel is ~5 Wt%.
6. Wash the heel slurry to lower the sodium concentration to ~0.5 M. Wash water is added to the PT at the same rate as the filter flux (4.6 gpm) to keep waste slurry at the PT process heel level. The heel and additional wash water are filtered similarly to the concentration step. Assuming that the filter is also backpulsed with 140 gallons of process water twice during washing and each backpulse lasts 2 seconds.
7. After washing, transfer heel slurry to the LPPPPT at 100 gpm.
8. Perform filter cleaning in five addition+cleaning steps : (1) 500 gallons of 1.0 M caustic (3 hours), (2) 500 gallons of DI water (2 hours), (3) 500 gallons of 2 Wt% oxalic acid (2 hours), (4) 500 gallons of DI water (2 hours), and (5) 500 gallons of 1.0 M caustic (3 hours). The spent filter cleaning chemicals are fed to PT.
9. After filter cleaning, transfer the spent filter cleaning chemicals from PT to LPPPPT at 100 gpm.

ARPOperation Task

ACM tasks provide an effective structure to handle batch processes. Using tasks, the following actions can be easily implemented during the simulation: changing the value of some variables, writing messages, suspending the simulation, creating snapshots, invoking scripts etc.

The ARP model makes use of the ACM task features.

The *ARPOperation* task is created to model all events outlined in the ARP operational sequence. Table 1 shows the entire *ARPOperation* Task.

Table 1. ARPOperation Task

```

Task ARPOperation runs when Time == 0
TimeNext as RealParameter;
WasteVolume as RealParameter;
NaOH_MST_VolRatio as RealParameter;
Solids_Wtpct as RealParameter;
MST_Massconc as RealParameter;
WasteVolume: 15.141647; // 4000 gallons waste
NaOH_MST_VolRatio: ARP_FeedAdjust.NaOH_MST_VolRatio;
Solids_Wtpct: 5;
MST_Massconc: ARP_FeedAdjust.cMass_MST; // g/L

// Transfer 4000 gallons salt waste solution to PT
Print "***** ADD A NEW BATCH OF WASTE TO PT *****";
Waste.Feed_Signal: 1;
Waste.FV_Feed: 9.085; // 40 gpm
TimeNext: Time + (WasteVolume/Waste.FV_Feed);
Wait For Time == TimeNext;
Waste.Feed_Signal: 0;

// Add NaOH and MST to PT
Print "***** ADD 1.66M NAOH AND 15 Wt% MST TO PT *****";
NaOH.Feed_Signal: 1;
NaOH.FV_Feed: 9.085* NaOH_MST_VolRatio/(NaOH_MST_VolRatio
+ 1);
NaOH.conc("H2O"): 55.6;
NaOH.conc("NaOH"): 1.66;
MST.Feed_Signal: 1;
MST.FV_Feed: NaOH.FV_Feed / NaOH_MST_VolRatio;
TimeNext: Time + (ARP_FeedAdjust.NaOH_Volume/NaOH.FV_Feed);
Wait For Time == TimeNext;
NaOH.Feed_Signal: 0;
MST.Feed_Signal: 0;
TimeNext: Time + 24;

// 24 hours for MST to adsorb Sr/Actinides
Print "***** START 24 HOURS OF Sr/ACTINIDES ADSORPTION
*****";
ARP.MST_Adsorption: 1;
Wait For Time == TimeNext;
ARP.MST_Adsorption: 0;

// Concentrate the salt solution to heel (950 gallons)
// But first, add 280 gallons backpulse
Print "***** FILTER THE SALT SOLUTION TO HEEL *****";
Print "***** BUT FIRST, ADD 280 GALLONS WATER OF FILTER
BACKPULSE *****";
H2O.Feed_Signal: 1;
H2O.FV_Feed: 953.92378; // 4200 gpm
TimeNext: Time + (4/3600); // 4 seconds of backpulses
ARP.Filtrate_ON: 0;
Wait For Time == TimeNext;
H2O.Feed_Signal: 0;

// Start filtration
Print "***** START FILTRATION *****";
ARP.Heel_Volume: 3.5961412; // 950 gallons
ARP.Filtrate_ON: 1;
ARP.FV_Filtrate: 1.0447737; // 4.6 gpm
Wait For ARP.Volume <= ARP.Heel_Volume;
ARP.Filtrate_ON: 0;

```

```

// Check solids Wt% in heel solution
IF ARP.Solids_Wtpct < Solids_Wtpct THEN
  // Not enough solids, add another batch of salt waste solution
  Print "***** NOT ENOUGH SOLIDS IN HEEL ==> NEW BATCH
OF WASTE NEEDED! *****";
  ARP.BatchNum: ARP.BatchNum + 1;
  ARP_FeedAdjust.c_Na: 5.6;
  ARP_FeedAdjust.cMass_MST: 0.4;
  Restart When ARP.Solids_Wtpct < Solids_Wtpct;
ELSE
  // Enough solids, start washing the heel solution to ~0.5M sodium
  // But first, add 280 gallons backpulse
  Print "***** WASHING THE HEEL SOLUTION TO 0.5M
Na+*****";
  Print "***** BUT FIRST, ADD 280 GALLONS WATER OF FILTER
BACKPULSE *****";
  H2O.Feed_Signal: 1;
  H2O.FV_Feed: 953.92378; // 4200 gpm
  TimeNext: Time + (4/3600); // 4 seconds of backpulses
  ARP.Filtrate_ON: 0;
ENDIF
Wait For Time == TimeNext;
H2O.Feed_Signal: 0;
// Start washing
H2O.Feed_Signal: 1;
H2O.FV_Feed: 1.0447737; // 4.6 gpm
ARP.Filtrate_ON: 1;
ARP.FV_Filtrate: 1.0447737; // 4.6 gpm
Wait For ARP.c_Na <= 0.5;
H2O.Feed_Signal: 0;
ARP.Filtrate_ON: 0;

// Transfer heel to LPPPT
Print "***** TRANSFER PT HEEL TO LPPPT *****";
ARP.Heel_Volume: 1E-6;
ARP.SludgeTransfer_ON: 1;
ARP.FV_Sludge: 22.712471; // 100 gpm
Wait For ARP.Volume < 1E-6;
ARP.SludgeTransfer_ON: 0;

// Start filter cleaning with chemicals
Print "***** START FILTER CLEANING WITH CHEMICALS
*****";
// Filter cleaning with 500 gallons of 1M NaOH
Print "***** FILTER CLEANING WITH 500 GALLONS OF 1M
NAOH (STEP 1) ";
NaOH.Feed_Signal: 1;
NaOH.FV_Feed: 0.630902; // 500 gallons in 3 hours
NaOH.conc("H2O"): 55.6;
NaOH.conc("NaOH"): 1.0;
TimeNext: Time + 3;
Wait For Time == TimeNext;
NaOH.Feed_Signal: 0;
// Filter cleaning with 500 gallons of de-ionized water
Print "***** FILTER CLEANING WITH 500 GALLONS OF DI
WATER (STEP 2)";
H2O.Feed_Signal: 1;
H2O.FV_Feed: 0.9463529; // 500 gallons in 2 hours
TimeNext: Time + 2;
Wait For Time == TimeNext;
H2O.Feed_Signal: 0;
// Filter cleaning with 500 gallons of 2Wt% Oxalic Acid
Print "***** FILTER CLEANING WITH 500 GALLONS OF 2WT%
OXALIC ACID (STEP 3)";
OxalicAcid.Feed_Signal: 1;

```

```

OxalicAcid.FV_Feed: 0.9463529; // 500 gallons in 2 hours
ARP.k_OxalicAcid: 1000;
TimeNext: Time + 2;
Wait For Time == TimeNext;
OxalicAcid.Feed_Signal: 0;
// Filter cleaning with 500 gallons of de-ionized water
Print "***** FILTER CLEANING WITH 500 GALLONS OF DI
WATER (STEP 4)";
H2O.Feed_Signal: 1;
H2O.FV_Feed: 0.9463529; // 500 gallons in 2 hours
TimeNext: Time + 2;
Wait For Time == TimeNext;
H2O.Feed_Signal: 0;
// Filter cleaning with 500 gallons of 1M NaOH
Print "***** FILTER CLEANING WITH 500 GALLONS OF 1M
NAOH (STEP 5)";
NaOH.Feed_Signal: 1;
NaOH.FV_Feed: 0.630902; // 500 gallons in 3 hours
NaOH.conc("H2O"): 55.6;
NaOH.conc("NaOH"): 1.0;
TimeNext: Time + 3;
Wait For Time == TimeNext;
NaOH.Feed_Signal: 0;
ARP.k_OxalicAcid: 0;

// Transfer the spent filter cleaning chemicals from PT to LPPPT
Print "***** TRANSFER SPENT FILTER CLEANING CHEMICALS
FROM PT TO LP PPPT *****";
ARP.Heel_Volume: 1E-6;
ARP.SludgeTransfer_ON: 1;
ARP.FV_Sludge: 22.712471; // 100 gpm
Wait For ARP.Volume < 1E-6;
ARP.SludgeTransfer_ON: 0;

```

Pause;

End

Block Model Description

The ARP flowsheet shown in Figure 2 above consists of blocks and streams that are based on block and stream models. Table 2 lists all the blocks, streams and the underlying models. The two block models *PrecipitateTank* and *FeedAdjust* are discussed in the following subsections.

Table 2. List of Blocks, Streams and Models

| Block | Block Model |
|----------------|-----------------|
| ARP | PrecipitateTank |
| ARP_FeedAdjust | FeedAdjust |
| Stream | Stream Model |
| Waste | Feed |
| MST | Slurry |
| NAOH | Feed |
| H2O | Feed |
| OxalicAcid | Chemical |
| Filtrate | ACM Connection |
| Sludge | ACM Connection |

PrecipitateTank Model

This model describes both the PT and filter operations of the ARP. The following calculations are performed within the *PrecipitateTank* model:

1. Overall mole balance
2. Component mole balances including radioactive components
3. Adsorption of strontium and actinides on MST
4. Reaction of oxalic acid and caustic
5. Filtrate output stream to the FHT
6. MST/sludge output stream to the LPPPT

Adsorption of strontium and actinides by MST is assumed to proceed to equilibrium. Decontamination factors (DF) for the radionuclides are calculated from their respective equilibrium relationship or given K_d factors using


$$DF = \frac{C_o}{C_f} \quad \text{and} \quad K_d = (DF - 1) \frac{V}{m}$$

where C_o is the initial concentration of the adsorbed nuclides, C_f is the final concentration at equilibrium, m is the mass of MST added to the solution and V is the solution volume. Values of DFs used in the model are 100, 12 and 1.5 for strontium, plutonium and uranium, respectively. The adsorption rate R_a for each nuclide is calculate as follows:

$$R_a = k_a (CK_d - C_s)$$

k_a is rate constant, C is the concentration of the nuclides dissolved in the liquid phase and C_s is the concentration of the nuclides adsorbed on MST.

Oxalic acid will react with caustic to form sodium oxalate and water. The stoichiometry for this reaction is given as



The rate R of this reaction can be expressed by the following equation:

$$R = kVC_{\text{OxalicAcid}}C_{\text{Caustic}}^2$$

where k is rate constant, V is the solution volume and C is the concentration.

FeedAdjust Model

The *FeedAdjust* model calculates the mixing of a given waste volume of specified concentrations with caustic and

MST to satisfy the requirements of 5.6 M sodium and 0.4 g MST/L.

The calculations performed in the *FeedAdjust* model are:

1. Overall mole balance
2. Component mole balances
3. Required caustic volume
4. Required MST volume
5. Volume ratio of caustic to MST

The calculated caustic volume and caustic-to-MST volume ratio are used in the *ARPOperation* task.

A SAMPLE SIMULATION

For demonstration, a typical simulation of the ARP model is provided based on specification of an average waste type. The scenario presented here corresponds to the ARP operational sequence discussed above. The simulation results are shown in Figures 3 and 4. Figure 3 displays the concentrations of strontium, plutonium and uranium dissolved in the solution. Figure 4 exhibits major variables in the PT including MST concentration (g/L), sodium concentration (mol/L), solids Wt% and tank volume (gals). As expected based on specified DFs, the results show that the decontamination of strontium and plutonium is an order of magnitude more than that of uranium. Twenty six batches must be processed until the 5 Wt% solids concentration in the PT process heel is achieved. The total cycle time including the filter cleaning process is ~1176 hours.





Figure 3. Concentrations of dissolved radionuclides**Figure 4.** Precipitate tank

CONCLUSIONS

Using the Aspen Custom Modeler™ (ACM) software package, a dynamic model has been developed for the Actinide Removal Process (ARP) at the Savannah River Site. The model is capable of accurately representing all steps of the ARP batch operational sequence. The total cycle time to process a specific waste is computed. Concentrations of radionuclides remaining in the solution are predicted based on specified decontamination factors.

REFERENCES

Hang, T. and D.D. Walker, 2001. "Prediction of Gas Generation and Bubble Formation in Crystalline Silicotitanate Ion Exchange Columns." In *Proceedings of the 2001 Advanced Simulation Technologies Conference*

(Seattle, WA, April 22-26, 2001), The Society for Modeling and Simulation International, 63-68.

Hang, T., M.B. Gorenssek and L.D. Koffman, 2003. "Modeling of the Savannah River Site High Level Waste Evaporator Systems." In *Proceedings of the 2003 Advanced Simulation Technologies Conference* (Orlando, FL, March 30 - April 3, 2003), The Society for Modeling and Simulation International, 71-76.

Smith, F. G., III. "Dynamic Modeling of Radioactive Melter Off-Gas Systems." In *Proceedings of the 2003 Advanced Simulation Technologies Conference* (Orlando, FL, March 30 - April 3, 2003), The Society for Modeling and Simulation International, 57-61.

ACKNOWLEDGEMENT

The information contained in this paper was developed during the course of work done under Contract No. DE-AC0996SR18500 with the U.S. Department of Energy.